

Lab 11

X86-64

Questions from the Book

3.58 ♦

For a function with prototype

```
long decode2(long x, long y, long z);
```

gcc generates the following assembly code:

```
1  decode2:
2      subq    %rdx, %rsi
3      imulq   %rsi, %rdi
4      movq    %rsi, %rax
5      salq    $63, %rax
6      sarq    $63, %rax
7      xorq    %rdi, %rax
8      ret
```

Parameters `x`, `y`, and `z` are passed in registers `%rdi`, `%rsi`, and `%rdx`. The code stores the return value in register `%rax`.

Write C code for `decode2` that will have an effect equivalent to the assembly code shown.

3.60 ♦♦

Consider the following assembly code:

```
long loop(long x, int n)
x in %rdi, n in %esi
1  loop:
2      movl    %esi, %ecx
3      movl    $1, %edx
4      movl    $0, %eax
5      jmp     .L2
6  .L3:
7      movq    %rdi, %r8
8      andq    %rdx, %r8
9      orq     %r8, %rax
10     salq    %cl, %rdx
11  .L2:
12     testq   %rdx, %rdx
13     jne     .L3
14     rep; ret
```

The preceding code was generated by compiling C code that had the following overall form:

```
1  long loop(long x, long n)
2  {
3      long result = _____;
4      long mask;
5      for (mask = _____; mask _____ ; mask = _____ ) {
6          result |= _____ ;
7      }
8      return result;
9  }
```

Your task is to fill in the missing parts of the C code to get a program equivalent to the generated assembly code. Recall that the result of the function is returned in register `%rax`. You will find it helpful to examine the assembly code before, during, and after the loop to form a consistent mapping between the registers and the program variables.

- A. Which registers hold program values `x`, `n`, `result`, and `mask`?
- B. What are the initial values of `result` and `mask`?
- C. What is the test condition for `mask`?
- D. How does `mask` get updated?
- E. How does `result` get updated?
- F. Fill in all the missing parts of the C code.

3.65 ♦

The following code transposes the elements of an $M \times M$ array, where M is a constant defined by `#define`:

```
1 void transpose(long A[M][M]) {
2     long i, j;
3     for (i = 0; i < M; i++)
4         for (j = 0; j < i; j++) {
5             long t = A[i][j];
6             A[i][j] = A[j][i];
7             A[j][i] = t;
8         }
9 }
```

When compiled with optimization level `-O1`, gcc generates the following code for the inner loop of the function:

```
1 .L6:
2     movq    (%rdx), %rcx
3     movq    (%rax), %rsi
4     movq    %rsi, (%rdx)
5     movq    %rcx, (%rax)
6     addq    $8, %rdx
7     addq    $120, %rax
8     cmpq    %rdi, %rax
9     jne     .L6
```

We can see that gcc has converted the array indexing to pointer code.

- A. Which register holds a pointer to array element `A[i][j]`?
- B. Which register holds a pointer to array element `A[j][i]`?
- C. What is the value of M ?

3.66 ♦

Consider the following source code, where NR and NC are macro expressions declared with `#define` that compute the dimensions of array A in terms of parameter *n*. This code computes the sum of the elements of column *j* of the array.

```
1  long sum_col(long n, long A[NR(n)][NC(n)], long j) {
2      long i;
3      long result = 0;
4      for (i = 0; i < NR(n); i++)
5          result += A[i][j];
6      return result;
7  }
```

In compiling this program, GCC generates the following assembly code:

```
long sum_col(long n, long A[NR(n)][NC(n)], long j)
n in %rdi, A in %rsi, j in %rdx
1  sum_col:
2      leaq    1(,%rdi,4), %r8
3      leaq    (%rdi,%rdi,2), %rax
4      movq    %rax, %rdi
5      testq   %rax, %rax
6      jle     .L4
7      salq    $3, %r8
8      leaq    (%rsi,%rdx,8), %rcx
9      movl    $0, %eax
10     movl    $0, %edx
11  .L3:
12     addq    (%rcx), %rax
13     addq    $1, %rdx
14     addq    %r8, %rcx
15     cmpq    %rdi, %rdx
16     jne     .L3
17     rep; ret
18  .L4:
19     movl    $0, %eax
20     ret
```

Use your reverse engineering skills to determine the definitions of NR and NC.

3.67 ♦♦

For this exercise, we will examine the code generated by gcc for functions that have structures as arguments and return values, and from this see how these language features are typically implemented.

The following C code has a function `process` having structures as argument and return values, and a function `eval` that calls `process`:

```
1  typedef struct {
2      long a[2];
3      long *p;
4  } strA;
5
6  typedef struct {
7      long u[2];
8      long q;
9  } strB;
10
11 strB process(strA s) {
12     strB r;
13     r.u[0] = s.a[1];
14     r.u[1] = s.a[0];
15     r.q = *s.p;
16     return r;
17 }
18
19 long eval(long x, long y, long z) {
20     strA s;
21     s.a[0] = x;
22     s.a[1] = y;
23     s.p = &z;
24     strB r = process(s);
25     return r.u[0] + r.u[1] + r.q;
26 }
```

Continued on next page

```

    strB process(strA s)
1  process:
2      movq    %rdi, %rax
3      movq    24(%rsp), %rdx
4      movq    (%rdx), %rdx
5      movq    16(%rsp), %rcx
6      movq    %rcx, (%rdi)
7      movq    8(%rsp), %rcx
8      movq    %rcx, 8(%rdi)
9      movq    %rdx, 16(%rdi)
10     ret

    long eval(long x, long y, long z)
    x in %rdi, y in %rsi, z in %rdx
1  eval:
2      subq    $104, %rsp
3      movq    %rdx, 24(%rsp)
4      leaq    24(%rsp), %rax
5      movq    %rdi, (%rsp)
6      movq    %rsi, 8(%rsp)
7      movq    %rax, 16(%rsp)
8      leaq    64(%rsp), %rdi
9      call    process
10     movq    72(%rsp), %rax
11     addq    64(%rsp), %rax
12     addq    80(%rsp), %rax
13     addq    $104, %rsp
14     ret

```

- A. We can see on line 2 of function `eval` that it allocates 104 bytes on the stack. Diagram the stack frame for `eval`, showing the values that it stores on the stack prior to calling `process`.
- B. What value does `eval` pass in its call to `process`?
- C. How does the code for `process` access the elements of structure argument `s`?
- D. How does the code for `process` set the fields of result structure `r`?
- E. Complete your diagram of the stack frame for `eval`, showing how `eval` accesses the elements of structure `r` following the return from `process`.
- F. What general principles can you discern about how structure values are passed as function arguments and how they are returned as function results?

3.68 ♦♦♦

In the following code, *A* and *B* are constants defined with `#define`:

```
1  typedef struct {
2      int x[A][B]; /* Unknown constants A and B */
3      long y;
4  } str1;
5
6  typedef struct {
7      char array[B];
8      int t;
9      short s[A];
10     long u;
11 } str2;
12
13 void setVal(str1 *p, str2 *q) {
14     long v1 = q->t;
15     long v2 = q->u;
16     p->y = v1+v2;
17 }
```

Gcc generates the following code for `setVal`:

```
void setVal(str1 *p, str2 *q)
p in %rdi, q in %rsi
1  setVal:
2      movslq  8(%rsi), %rax
3      addq    32(%rsi), %rax
4      movq    %rax, 184(%rdi)
5      ret
```

What are the values of *A* and *B*? (The solution is unique.)

EXTRA CREDIT

3.64 ♦♦♦

Consider the following source code, where R , S , and T are constants declared with `#define`:

```
1  long A[R][S][T];
2
3  long store_ele(long i, long j, long k, long *dest)
4  {
5      *dest = A[i][j][k];
6      return sizeof(A);
7  }
```

In compiling this program, GCC generates the following assembly code:

```
long store_ele(long i, long j, long k, long *dest)
i in %rdi, j in %rsi, k in %rdx, dest in %rcx
1  store_ele:
2      leaq    (%rsi,%rsi,2), %rax
3      leaq    (%rsi,%rax,4), %rax
4      movq    %rdi, %rsi
5      salq    $6, %rsi
6      addq    %rsi, %rdi
7      addq    %rax, %rdi
8      addq    %rdi, %rdx
9      movq    A(,%rdx,8), %rax
10     movq    %rax, (%rcx)
11     movl    $3640, %eax
12     ret
```

- A. Extend **Equation 3.1** from two dimensions to three to provide a formula for the location of array element $A[i][j][k]$.
- B. Use your reverse engineering skills to determine the values of R , S , and T based on the assembly code.